
/*******************HW11Cartoon.sas ******************/
title 'STA441s16 HW 11 Check';
proc format;  /* value labels used in data step below */
     value cfmt 0 = 'Black & White'     1 = 'Colour';
     value efmt 0 = 'Pre-professional'  1 = 'Professional'  2 = 'Student';
     value lfmt 1 = 'Hospital A '  2 = 'Hospital B'
                3 = 'Hospital C' 4 = 'Penn State';
data Walt;
     infile '/folders/myfolders/cartoon.data.txt';
     input id colour educ location otis cartoon1 real1 cartoon2 real2;
     label
       colour    = 'Colour versus Black & white'
       educ      = 'Education'
       location  = 'Where did respondent come from?'
       otis      = 'Otis Mental Ability Test'
       cartoon1  = 'Cartoon test score at time 1'
       real1     = 'Realistic test score at time 1'
       cartoon2  = 'Cartoon test score at time 2'
       real2     = 'Realistic test score at time 2'
       memory    = 'Recall of training materials'
       type      = 'Training materials: Cartoon versus realistic ';
     format colour cfmt.;
     format educ efmt.;
     format location lfmt.; 
proc freq;
     tables location*educ  / norow nocol nopercent missing;   

data disney;
     set walt;
     if educ=2; /* Just the students */
     time = 1; type = 'Real';     memory = real1; output;
     time = 1; type = 'Cartoon';  memory = cartoon1; output;
     time = 2; type = 'Real';     memory = real2; output;
     time = 2; type = 'Cartoon';  memory = cartoon2; output;
     keep  id otis colour time type memory; 
     
/* There were 70 students from Penn State. */

proc freq;
     tables time*colour*type / norow nocol nopercent missing;

proc print data=disney(obs=16); 
     var id otis colour time type memory;

proc mixed;
     title2 'Proc Mixed repeated measures';
     class colour time type;
     model memory = otis time|colour|type;
     repeated / type=un subject=id r;
     lsmeans time type colour*time;  /* Look at significant effects. */

